In this chapter, we will study the value of European digital and share digital options and standard European puts and calls under the Black-Scholes assumptions. We will also explain how to calculate implied volatilities and the option "Greeks." The Black-Scholes assumptions are that the underlying asset pays a constant dividend yield q and has price S satisfying

$$\frac{\mathrm{d}S}{S} = \mu \,\mathrm{d}t + \sigma \,\mathrm{d}B \tag{3.1}$$

for a Brownian motion B. Here σ is assumed to be constant (though we will allow it to vary in a non-random way at the end of the chapter) and μ can be a quite general random process. It is also assumed that there is a constant continuously-compounded risk-free rate r.

Under these assumptions, we will complete the discussion of Sect. 1.5 to derive option pricing formulas. Recall that, to price a European call option, all that remains to be done is to calculate the probabilities of the option finishing in the money when we use the risk-free asset and the underlying asset as numeraires. We will do this using the results of Sect. 2.9. As in Sect. 1.5, we will approach the pricing of call and put options by first considering their basic building blocks: digitals and share digitals.

3.1 Digital Options

A digital (or "binary") option pays a fixed amount in a certain event and zero otherwise. Consider a digital that pays \$1 at date T if S(T) > K, where K is a number that is fixed by the contract. This means that the digital pays x dollars at date T where x is defined as

$$x = \begin{cases} 1 & \text{if } S(T) > K ,\\ 0 & \text{otherwise }. \end{cases}$$

3

Using the risk-neutral pricing formula (1.18), the value of the digital at date 0 is $e^{-rT}E^R[x]$. Note that

$$E^{R}[x] = 1 \times \operatorname{prob}^{R}(x=1) + 0 \times \operatorname{prob}^{R}(x=0)$$

= $\operatorname{prob}^{R}(x=1)$
= $\operatorname{prob}^{R}(S(T) > K)$.

So we need to calculate this probability of the digital finishing in the money.

In Sect. 2.9—see (2.27)—we learned that under the Black-Scholes assumption (3.1) we have

$$\frac{\mathrm{d}S}{S} = (r-q)\,\mathrm{d}t + \sigma\,\mathrm{d}B^*\;,$$

where B^* is a Brownian motion under the risk-neutral measure.¹ In Sect. 2.8, we observed that this is equivalent to

$$d\log S = \left(r - q - \frac{1}{2}\sigma^2\right) dt + \sigma dB^*$$
.

Now using the formulas (2.34)–(2.35), with $\alpha = r - q - \sigma^2/2$, we have prob^{*R*} $(S(T) > K) = N(d_2)$ where

$$d_2 = \frac{\log\left(\frac{S(0)}{K}\right) + \left(r - q - \frac{1}{2}\sigma^2\right)T}{\sigma\sqrt{T}} . \tag{3.2}$$

The notation d_2 is standard notation from the Black-Scholes formula, and we use it—rather than a simple d—to distinguish the number (3.2) from a similar number—to be called d_1 of course—that we will see in the next section. We conclude:

The value of a digital option that pays \$1 when S(T) > K is $e^{-rT} N(d_2)$, where d_2 is defined in (3.2).

Consider now a digital that pays when the underlying asset price is low; i.e., consider a security that pays y dollars at date T where

$$y = \begin{cases} 1 & \text{if } S(T) < K \\ 0 & \text{otherwise .} \end{cases}$$

Using risk-neutral pricing again, the value of this digital at date 0 is

¹ There is no other risky asset price Y in this model, so the subscripts we used in Sect. 2.9 on the volatility coefficients and on B and B^* to distinguish the Brownian motion driving S from the Brownian motion driving Y and to distinguish their volatilities are not needed here.

3.2 Share Digitals 51

$$e^{-rT}E^{R}[y] = e^{-rT}prob^{R}(y=1) = e^{-rT}prob^{R}(S(T) < K)$$
.

From this fact and the formula (2.36), we conclude:

The value of a digital option that pays \$1 when S(T) < K is $e^{-rT} N(-d_2)$, where d_2 is defined in (3.2).

3.2 Share Digitals

Consider a derivative security that pays one share of the underlying asset at date T if S(T) > K and pays zero otherwise. This is called a "share digital." As before, let

$$x = \begin{cases} 1 & \text{if } S(T) > K ,\\ 0 & \text{otherwise }. \end{cases}$$

Then the payoff of the share digital at date T is xS(T). Let Y(t) denote the value of this claim for $0 \le t \le T$. We have Y(T) = xS(T) and we want to find Y(0).

From Sect. 2.7, we know that $V(t) = e^{qt}S(t)$ is the price of a non-dividendpaying portfolio. From our fundamental pricing formula (1.17), using V as the numeraire, we have

$$Y(0) = S(0)E^V \left[\frac{Y(T)}{e^{qT}S(T)}\right]$$
$$= e^{-qT}S(0)E^V[x] .$$

As in the previous section, $E^{V}[x] = \text{prob}^{V}(x = 1)$, so we need to compute this probability of the option finishing in the money.

We follow the same steps as in the previous section. From (2.28) we have

$$\frac{\mathrm{d}S}{S} = (r - q + \sigma^2) \,\mathrm{d}t + \sigma \,\mathrm{d}B^*,$$

where now B^* denotes a Brownian motion when V is the numeraire. This is equivalent to

$$d\log S = \left(r - q + \frac{1}{2}\sigma^2\right) dt + \sigma dB^* .$$
(3.3)

Thus, from the formulas (2.34)–(2.35), with $\alpha = r - q + \sigma^2/2$, we have

$$\operatorname{prob}^{V}(S(T) > K) = \mathcal{N}(d_{1}) ,$$

where

$$d_1 = \frac{\log\left(\frac{S(0)}{K}\right) + \left(r - q + \frac{1}{2}\sigma^2\right)T}{\sigma\sqrt{T}} .$$
(3.4)

This implies:

The value of a share digital that pays one share when S(T) > K is $e^{-qT}S(0) N(d_1)$, where d_1 is defined in (3.4).

Consider now a share digital that pays one share of the stock at date T if S(T) < K. Letting

$$y = \begin{cases} 1 & \text{if } S(T) < K \\ 0 & \text{otherwise }, \end{cases}$$

the payoff of this option is yS(T). Its value at date 0 is

$$e^{-qT}S(0)E^{V}[y] = e^{-qT}S(0) \times \operatorname{prob}^{V}(y=1)$$
$$= e^{-qT}S(0) \times \operatorname{prob}^{V}(S(T) < K),$$

and from the formula (2.36) we have

$$\operatorname{prob}^{V}(S(T) < K) = \operatorname{N}(-d_1)$$

We conclude:

The value of a share digital that pays one share when S(T) < K is $e^{-qT}S(0) N(-d_1)$, where d_1 is defined in (3.4).

3.3 Puts and Calls

A European call option pays S(T) - K at date T if S(T) > K and 0 otherwise. Again letting

$$x = \begin{cases} 1 & \text{if } S(T) > K \\ 0 & \text{otherwise }, \end{cases}$$

the payoff of the call can be written as xS(T) - xK. This is equivalent to one share digital minus K digitals, with the digitals paying in the event that S(T) > K. The share digital is worth $e^{-qT}S(0) N(d_1)$ at date 0 and each digital is worth $e^{-rT} N(d_2)$. Note that equations (3.2) and (3.4) for d_1 and d_2 imply $d_2 = d_1 - \sigma \sqrt{T}$. Therefore, combining the results of the previous two sections yields the Black-Scholes formula:

The value of a European call option at date 0 is

$$e^{-qT}S(0) N(d_1) - e^{-rT}K N(d_2)$$
, (3.5)
where d_1 is defined in (3.4) and $d_2 = d_1 - \sigma \sqrt{T}$.

A European put option pays K - S(T) at date T if S(T) < K and 0 otherwise. As before, let

$$y = \begin{cases} 1 & \text{if } S(T) < K \\ 0 & \text{otherwise .} \end{cases}$$

The payoff of the put option is yK - yS(T). This is equivalent to K digitals minus one share digital, all of the digitals paying when S(T) < K. Thus, we have:

The value of a European put option at date 0 is

$$e^{-rT}KN(-d_2) - e^{-qT}S(0)N(-d_1)$$
, (3.6)
where d_1 is defined in (3.4) and $d_2 = d_1 - \sigma\sqrt{T}$.

Again, this is the Black-Scholes formula.

The values of the European put and call satisfy put-call parity, and we can also find one from the other by^2

$$e^{-rT}K + Call Price = e^{-qT}S(0) + Put Price$$
. (3.7)

3.4 Greeks

The derivatives (calculus derivatives, not financial derivatives!) of an option pricing formula with respect to the inputs are commonly called "Greeks." The most important Greek is the option "delta." This measures the sensitivity of the option value to changes in the value of the underlying asset. The following table shows the standard Greeks, with reference to the Black-Scholes pricing formula.

² The put-call parity relation follows from the fact that both the left and the righthand sides are the prices of portfolios that have value $\max(S(T), K)$ at the maturity of the option. To see this for the left-hand side, note that $e^{-rT}K$ is sufficient cash to accumulate to K at date T, allowing exercise of the call when it is in the money and retention of the cash K otherwise. For the right-hand side, note that $e^{-qT}S(0)$ is enough cash to buy e^{-qT} shares of the stock at date 0 which, with reinvestment of dividends, will accumulate to one share at date T, enabling exercise of the put if it is in the money or retention of the share otherwise.

Input	Input Symbol	Greek	Greek Symbol
Stock price	S	delta	δ
delta	δ	gamma	Г
- Time to maturity	-T	theta	Θ
Volatility	σ	vega	\mathcal{V}
Interest rate	r	rho	ρ

Table 3.1. Black-Scholes Greeks

The second line of the above shows δ as an input.³ Of course, it is not an input but instead is calculated. Gamma, the derivative of δ , is the second derivative of the option price with respect to the underlying asset price. The reason for calculating Θ as the derivative with respect to -T instead of T is that the time-to-maturity T decreasing (-T increasing) is equivalent to time passing, so Θ measures the change in the option value when time passes.

We can calculate these from the Black-Scholes formula using the chain rule from differential calculus. The derivative of the normal distribution function N is the normal density function n defined as

$$n(d) = \frac{1}{\sqrt{2\pi}} e^{-d^2/2}$$
.

One can easily verify directly that

$$e^{-qT}Sn(d_1) = e^{-rT}Kn(d_2)$$
, (3.8)

which simplifies the calculations for the Black-Scholes call option pricing formula. For this formula, the Greeks are as follows:

$$\begin{split} \delta &= \mathrm{e}^{-qT} \,\mathrm{N}(d_1) + \mathrm{e}^{-qT} S \,\mathrm{n}(d_1) \frac{\partial d_1}{\partial S} - \mathrm{e}^{-rT} K \,\mathrm{n}(d_2) \frac{\partial d_2}{\partial S} \\ &= \mathrm{e}^{-qT} \,\mathrm{N}(d_1) + \mathrm{e}^{-qT} S \,\mathrm{n}(d_1) \left(\frac{\partial d_1}{\partial S} - \frac{\partial d_2}{\partial S} \right) \\ &= \mathrm{e}^{-qT} \,\mathrm{N}(d_1) \;, \\ \Gamma &= \mathrm{e}^{-qT} \,\mathrm{n}(d_1) \frac{\partial d_1}{\partial S} = \mathrm{e}^{-qT} \,\mathrm{n}(d_1) \frac{1}{S\sigma\sqrt{T}} \;, \end{split}$$

³ The delta is frequently denoted by the upper case Δ , but we will use the lower case, reserving the upper case for discrete changes, e.g., Δt . One may have noticed also that the symbol for vega is a little different from the others; this reflects the fact that vega is not actually a Greek letter.

3.5 Delta Hedging 55

$$\begin{split} \Theta &= -\mathrm{e}^{-qT} S\,\mathrm{n}(d_1) \frac{\partial d_1}{\partial T} + q \mathrm{e}^{-qT} S\,\mathrm{N}(d_1) \\ &+ \mathrm{e}^{-rT} K\,\mathrm{n}(d_2) \frac{\partial d_2}{\partial T} - r \mathrm{e}^{-rT} K\,\mathrm{N}(d_2) \\ &= \mathrm{e}^{-qT} S\,\mathrm{n}(d_1) \left(\frac{\partial d_2}{\partial T} - \frac{\partial d_1}{\partial T} \right) \\ &+ q \mathrm{e}^{-qT} S\,\mathrm{N}(d_1) - r \mathrm{e}^{-rT} K\,\mathrm{N}(d_2) \\ &= -\mathrm{e}^{-qT} S\,\mathrm{n}(d_1) \frac{\sigma}{2\sqrt{T}} + q \mathrm{e}^{-qT} S\,\mathrm{N}(d_1) - r \mathrm{e}^{-rT} K\,\mathrm{N}(d_2) \,, \\ \mathcal{V} &= \mathrm{e}^{-qT} S\,\mathrm{n}(d_1) \frac{\partial d_1}{\partial \sigma} - \mathrm{e}^{-rT} K\,\mathrm{n}(d_2) \frac{\partial d_2}{\partial \sigma} \\ &= \mathrm{e}^{-qT} S\,\mathrm{n}(d_1) \left(\frac{\partial d_1}{\partial \sigma} - \frac{\partial d_2}{\partial \sigma} \right) \\ &= \mathrm{e}^{-qT} S\,\mathrm{n}(d_1) \sqrt{T} \,, \\ \rho &= \mathrm{e}^{-qT} S\,\mathrm{n}(d_1) \frac{\partial d_1}{\partial r} - \mathrm{e}^{-rT} K\,\mathrm{n}(d_2) \frac{\partial d_2}{\partial r} + T \mathrm{e}^{-rT} K\,\mathrm{N}(d_2) \\ &= \mathrm{e}^{-qT} S\,\mathrm{n}(d_1) \left(\frac{\partial d_1}{\partial r} - \frac{\partial d_2}{\partial \sigma} \right) + T \mathrm{e}^{-rT} K\,\mathrm{N}(d_2) \\ &= \mathrm{e}^{-qT} S\,\mathrm{n}(d_1) \left(\frac{\partial d_1}{\partial r} - \frac{\partial d_2}{\partial r} \right) + T \mathrm{e}^{-rT} K\,\mathrm{N}(d_2) \\ &= \mathrm{e}^{-qT} K\,\mathrm{N}(d_2) \,. \end{split}$$

We can calculate the Greeks of a European put option from the call option Greeks and put-call parity:

Put Price = Call Price +
$$e^{-rT}K - e^{-qT}S(0)$$
.

For example, the delta of a put is the delta of a call (with the same strike and maturity) minus e^{-qT} , and the gamma of a put is the same as the gamma of the corresponding call.

3.5 Delta Hedging

The ability to create a fully hedged (risk-free) portfolio of the stock and an option is the essence of the arbitrage argument underlying the Black-Scholes formula, as we saw in Chap. 1 for the binomial model. For a call option, such a portfolio consists of delta shares of the underlying asset and a short call option, or a short position of delta shares of the underlying and a long call option. These portfolios have no instantaneous exposure to the price of the underlying. To create a perfect hedge, the portfolio must be adjusted continuously, because the delta changes when the price of the underlying changes and when time passes. In practice, any hedge will therefore be imperfect, even if the assumptions of the model are satisfied.

We first consider the continuous-time hedging argument. Consider a European call option with maturity T, and let C(S,t) denote the value of the

option at date t < T when the stock price is S at date t. Consider a portfolio that is short one call option and long δ shares of the underlying asset and that has a (short) cash position equal to $C - \delta S$. This portfolio has zero value at date t.

The change in the value of the portfolio in an instant dt is

$$-\mathrm{d}C + \delta\,\mathrm{d}S + q\delta S\,\mathrm{d}t + (C - \delta S)r\,\mathrm{d}t\,. \tag{3.9}$$

The first term reflects the change in the value of the option, the second term is the capital gain or loss on δ shares of stock, the third term is the dividends received on δ shares of stock, and the fourth term is the interest expense on the short cash position.

On the other hand, we know from Itô's formula that

$$dC = \frac{\partial C}{\partial S} dS + \frac{\partial C}{\partial t} dt + \frac{1}{2} \frac{\partial^2 C}{\partial S^2} (dS)^2$$
$$= \delta dS + \Theta dt + \frac{1}{2} \Gamma \sigma^2 S^2 dt . \qquad (3.10)$$

Substituting (3.10) into (3.9) shows that the change in the value of the portfolio is

$$-\Theta \,\mathrm{d}t - \frac{1}{2}\Gamma\sigma^2 S^2 \,\mathrm{d}t + q\delta S \,\mathrm{d}t + (C - \delta S)r \,\mathrm{d}t \,. \tag{3.11}$$

Several aspects of this are noteworthy. First, as noted earlier, the delta hedge (being long δ shares of the underlying) eliminates the exposure to changes in the price of the underlying—there is no dS term in (3.11). Second, Θ will be negative, because it captures the time decay in the option value; being short the option means the portfolio will profit from time decay at rate $-\Theta$. Third, this portfolio is "short gamma." We can also say it is "short convexity," the term "convexity" referring to the convex shape of the option value as a function of the price of the underlying, which translates mathematically to a positive second derivative (gamma). The volatility in the stock makes convexity valuable, and a portfolio that is short convexity will suffer losses. Finally, the portfolio is earning dividends but paying interest.

It is straightforward to check, from the definitions of Θ , Γ and δ in the preceding section, that the sum of the terms in (3.11) is zero. The time decay in the option value and dividends received on the shares of the underlying exactly offset the losses due to convexity and interest. Therefore, the delta hedge is a perfect hedge. The portfolio, which has a zero cost, neither earns nor loses money. This is true not only on average but for every possible change in the stock price.

To see how well this works with only discrete adjustments to the hedge, one can simulate the changes in S over time and sum the gains and losses over discrete rebalancing periods. One should input the actual (not risk-neutral) expected rate of return on the asset to compute the actual distribution of gains and losses. This is discussed further in Sect. 3.10.

3.6 Gamma Hedging

To attempt to improve the performance of a discretely rebalanced delta hedge, one can use another option to create a portfolio that is both delta and gamma neutral. Being delta neutral means hedged as in the previous section—the portfolio value has no exposure to changes in the underlying asset price. In other words, it means that the derivative of the portfolio value with respect to the price of the underlying (the portfolio delta) is zero. Being gamma neutral means that the delta of the portfolio has no exposure to changes in the underlying price, which is equivalent to the second derivative of the portfolio value with respect to the price of the underlying (the portfolio gamma) being zero. If the delta truly did not change, then there would be no need to rebalance continuously, and hence no hedging error introduced by only adjusting the portfolio at discrete times rather than continuously. However, there is certainly no guarantee that a discretely-rebalanced delta/gamma hedge will perform better than a discretely rebalanced delta hedge.

A delta/gamma hedge can be constructed as follows. Suppose we have written (shorted) a call option and we want to hedge both the delta and gamma using the underlying asset and another option, for example, another call option with a different strike. In practice, one would want to use a liquid option for this purpose, which typically means that the strike of the option will be near the current value of the underlying (i.e., the option used to hedge would be approximately at the money).

Let δ and Γ denote the delta and gamma of the written option and let δ' and Γ' denote the delta and gamma of the option used to hedge. Consider holding *a* shares of the stock and *b* units of the option used to hedge in conjunction with the short option. The delta of the stock is one (dS/dS = 1), so to obtain a zero portfolio delta we need

$$0 = -\delta + a + b\delta'. \tag{3.12}$$

The gamma of the stock is zero (d²S/dS² = $d \, 1/dS = 0$), so to obtain a zero portfolio gamma we need

$$0 = -\Gamma + b\Gamma' . aga{3.13}$$

Equation (3.13) shows that we should hold enough of the second option to neutralize the gamma of the option we have shorted; i.e.,

$$b = \frac{\Gamma}{\Gamma'}$$

Equation (3.12) shows that we should use the stock to delta hedge the portfolio of options; i.e.,

$$a = \delta - \frac{\Gamma}{\Gamma'} \delta' \; .$$

3.7 Implied Volatilities

All of the inputs into the option pricing formulas are in theory observable, except for the volatility coefficient σ . We can estimate σ from historical data (see Chap. 4), or estimate it from the prices of other options. The latter method exploits the fact that there is a one-to-one relationship between the price given by the Black-Scholes formula and the σ that is input, so one can take the price as given and infer σ from the formula. The σ computed in this way is called the "implied volatility." The implied volatility from one option can be used to price another (perhaps non-traded or less actively traded) option. The calculation of implied volatilities is discussed in Sect. 3.10.

Even if we acknowledge that the model is not correct, the computation of implied volatilities is still useful for characterizing market prices, because we can quickly describe an option as "expensive" or "cheap" depending on whether its implied volatility is large or small. Somewhat paradoxically, it is less easy to see if an option is expensive or cheap by looking at its price, because one must consider the price in the context of the exercise price and maturity. To some extent, the implied volatility normalizes the price relative to the exercise price and maturity. Of course, it does not always pay to sell expensive options or buy cheap options, unless they are expensive or cheap relative to an accurate model!

3.8 Term Structure of Volatility

The option pricing formulas in this chapter are derived from the fact that the natural logarithm of the stock price at maturity is normally distributed with a certain mean (depending on the numeraire) and variance equal to $\sigma^2 T$. It is not actually necessary that the volatility be constant. The formulas are still valid if

$$\frac{\mathrm{d}S(t)}{S(t)} = \mu(t)\,\mathrm{d}t + \sigma(t)\,\mathrm{d}B(t)$$

where $\sigma(t)$ is some non-random function of time (and again μ can be a quite general random process). In this case, the variance of log S(T) will be

$$\int_0^T \sigma^2(t) \,\mathrm{d}t \;, \tag{3.14}$$

which is essentially the sum of the instantaneous variances $\sigma^2(t) dt$. In the d_1 's and d_2 's in the option pricing formulas, $\sigma^2 T$ should be replaced by (3.14). A convenient way of expressing this is as follows. Let σ_{avg} be the positive number such that

$$\sigma_{\rm avg}^2 = \frac{1}{T} \int_0^T \sigma^2(t) \, \mathrm{d}t \;. \tag{3.15}$$

Then we simply need to input σ_{avg} as sigma in our option pricing functions. We will call σ_{avg} the "average volatility," though note that it is not really the average of $\sigma(t)$ but instead is the square root of the average of $\sigma^2(t)$.

It is important to recognize that, throughout this chapter, date 0 means the date at which the option is being valued. It is not necessarily the date at which the option was first bought or sold. So σ_{avg} is the average (in a sense) volatility during the remaining lifetime of the option, which need not be the same as the average during the option's entire lifetime. It is this remaining volatility that is important for pricing and hedging. Moreover, it is a mistake at date 0 to use $\sigma(0)$ as the volatility to compute prices and hedges. Instead, prices and hedges should be based on σ_{avg} .

These considerations provide a way to address the following situation. If we compute implied volatilities for options with different maturities, we will normally get different numbers. For example, consider two at-the-money options with maturities T_1 and T_2 where $T_2 > T_1$. Denote the implied volatilities by $\hat{\sigma}_1$ and $\hat{\sigma}_2$. We want to interpret these as average volatilities for the time periods $[0, T_1]$ and $[0, T_2]$ respectively. This requires the existence of a function $\sigma(t)$ such that

$$\hat{\sigma}_1^2 = \frac{1}{T_1} \int_0^{T_1} \sigma^2(t) \, \mathrm{d}t \quad \text{and} \quad \hat{\sigma}_2^2 = \frac{1}{T_2} \int_0^{T_2} \sigma^2(t) \, \mathrm{d}t$$

This would imply

$$\hat{\sigma}_2^2 T_2 - \hat{\sigma}_1^2 T_1 = \int_{T_1}^{T_2} \sigma^2(t) \,\mathrm{d}t \;,$$

which requires

$$\hat{\sigma}_2^2 T_2 - \hat{\sigma}_1^2 T_1 \ge 0$$

Equivalently,

$$\hat{\sigma}_2 \ge \sqrt{\frac{T_1}{T_2}} \hat{\sigma}_1$$

Provided this last inequality is satisfied, we can easily construct the function $\sigma(t)$ as

$$\sigma(t) = \begin{cases} \hat{\sigma}_1 & \text{for } t \le T_1 \\ \sqrt{\frac{\hat{\sigma}_2^2 T_2 - \hat{\sigma}_1^2 T_1}{T_2 - T_1}} & \text{for } T_1 < t \le T_2 \end{cases}$$

More generally, given a sequence of at-the-money options with maturities $T_1 < T_2 < \cdots T_N$ and implied volatilities $\hat{\sigma}_1, \ldots, \hat{\sigma}_N$, we define

$$\sigma(t) = \sqrt{\frac{\hat{\sigma}_{i+1}^2 T_{i+1} - \hat{\sigma}_i^2 T_i}{T_{i+1} - T_i}}$$

for $T_i < t \leq T_{i+1}$, provided the expression inside the square root symbol is positive. This $\sigma(t)$ is often called the "term structure of (implied) volatilities." Generally, we may expect $\sigma(t)$ to be a decreasing function of time t when the current market is especially volatile and to be an increasing function when the current market is especially quiet.

3.9 Smiles and Smirks

If we compute implied volatilities for options with the same maturity but different strikes, we will again obtain different implied volatilities for different options. If we plot implied volatility against the strike, the pattern one normally sees for equities and equity indices is the implied volatility declining as the strike increases until the strike is somewhere near the current value of the underlying (so the option is at the money). The implied volatility will then generally flatten out or increase slightly at higher strikes. The graph looks like a twisted smile (smirk). This pattern has been very pronounced in equity index option prices since the crash of 1987. In contrast to the term structure of implied volatilities, this "moneyness" structure of implied volatilities is simply inconsistent with the model. It suggests that the risk-neutral return distribution is not lognormal but instead exhibits a higher likelihood of extreme returns than the lognormal distribution (i.e., it has "fat tails") with the likelihood of extreme negative returns being higher than the likelihood of extreme positive returns (i.e., it is "skewed"). We will return to this subject in Sect. 4.6.

3.10 Calculations in VBA

The Black-Scholes call and put formulas and Greeks can easily be calculated in an Excel worksheet, using the standard functions Exp, Ln and the cumulative normal distribution function, which is provided in Excel as NormSDist. However, if these are to be used repeatedly, it is useful to create functions in VBA. In VBA, the cumulative normal distribution function is called Application.NormSDist. Also, the natural logarithm function in VBA is Log rather than Ln and the square root function in VBA is Sqr rather than Sqrt.

Black-Scholes Call and Put Formulas

The following function implements the Black-Scholes call pricing formula. For the sake of completeness, the function returns a value even when a volatility of zero is input, in which case the formula (3.5) is invalid (it involves division by zero in the calculation of d_1 and d_2). If the volatility is zero, then the stock is riskless and should appreciate at rate r - q. Moreover the option is riskless and its date–0 value should be the date–T value discounted at the risk-free rate. This implies that the call value at date 0 is⁴

⁴ This result can be verified by a simple arbitrage argument. For example, if the call value were less than this formula, then put-call parity would show that the put price is negative, which is impossible. On the other hand, if the call price is greater than this formula (and hence positive), then put-call parity shows that the put price is positive, and it is impossible that both the put and call will finish in the money (so, given that they are riskless, only one should have a positive value).

3.10 Calculations in VBA 61

$$e^{-rT} \max\left(0, e^{(r-q)T}S(0) - K\right) = \max\left(0, e^{-qT}S(0) - e^{-rT}K\right)$$
.

```
Function Black_Scholes_Call(S, K, r, sigma, q, T)
' Inputs are S = initial stock price
            K = strike price
            r = risk-free rate
            sigma = volatility
             q = dividend yield
             T = time to maturity
Dim d1, d2, N1, N2
If sigma = 0 Then
    Black_Scholes_Call = Application.Max(0,Exp(-q*T)*S-Exp(-r*T)*K)
Else
    d1 = (Log(S/K) + (r-q+0.5*sigma*sigma)*T) / (sigma*Sqr(T))
    d2 = d1 - sigma * Sqr(T)
    N1 = Application.NormSDist(d1)
    N2 = Application.NormSDist(d2)
    Black_Scholes_Call = Exp(-q*T)*S*N1 - Exp(-r*T)*K*N2
End If
End Function
```

It is useful to note that

Black_Scholes_Call(S,K,r,sigma,q,T)

gives the same result as

```
Black_Scholes_Call(exp(-q*T)*S,K,r,sigma,0,T).
```

In the latter formulation, we view the underlying asset as the portfolio which starts with e^{-qT} shares of the asset and reinvests dividends until date T. This portfolio has value S(T) at date T, so a European call option on this nondividend-paying portfolio is equivalent to a European call option on the stock. The initial value of the portfolio is $e^{-qT}S(0)$, which is input as the asset price in the latter formulation.

The Black-Scholes formula for the value of a European put option can be implemented as follows.

```
Function Black_Scholes_Put(S, K, r, sigma, q, T)
,
' Inputs are S = initial stock price
' K = strike price
' r = risk-free rate
' sigma = volatility
' q = dividend yield
' T = time to maturity
'
```

Dim d1, d2, N1, N2

```
If sigma = 0 Then
    Black_Scholes_Put = Application.Max(0,Exp(-r*T)*K-Exp(-q*T)*S)
Else
    d1 = (Log(S/K) + (r-q+0.5*sigma*sigma)*T) / (sigma*Sqr(T))
    d2 = d1 - sigma * Sqr(T)
    N1 = Application.NormSDist(-d1)
    N2 = Application.NormSDist(-d2)
    Black_Scholes_Put = Exp(-r*T)*K*N2 - Exp(-q*T)*S*N1
End If
End Function
```

Black-Scholes Greeks

The delta and gamma of a European call option can be computed with the following functions. The other Greeks are obviously calculated in a similar manner. Note that the constant $\pi = 3.14159...$ is provided in Excel as the "function" Pi() and can be accessed in Excel VBA as Application.Pi.

```
Function Black_Scholes_Call_Delta(S, K, r, sigma, q, T)
 Inputs are S = initial stock price
,
             K = strike price
             r = risk-free rate
             sigma = volatility
             q = dividend yield
             T = time to maturity
Dim d1, d2, N1, N2
d1 = (Log(S/K) + (r-q+0.5*sigma*sigma)*T) / (sigma*Sqr(T))
d2 = d1 - sigma * Sqr(T)
N1 = Application.NormSDist(d1)
N2 = Application.NormSDist(d2)
Black_Scholes_Call_Delta = Exp(-q*T)*N1
End Function
Function Black_Scholes_Call_Gamma(S, K, r, sigma, q, T)
' Inputs are S = initial stock price
             K = strike price
             r = risk-free rate
             sigma = volatility
,
             q = dividend yield
             T = time to maturity
Dim d1, d2, N1, N2, nd1
d1 = (Log(S/K) + (r-q+0.5*sigma*sigma)*T) / (sigma*Sqr(T))
d2 = d1 - sigma * Sqr(T)
N1 = Application.NormSDist(d1)
N2 = Application.NormSDist(d2)
```

```
nd1 = Exp(-d1 * d1 / 2) / Sqr(2 * Application.Pi)
Black_Scholes_Call_Gamma = Exp(-q*T)*nd1/(S*sigma*Sqr(T))
End Function
```

Implied Volatilities

We could find an implied volatility using the Solver tool, but then we would have to re-run Solver each time we changed one of the input values. We will need to solve similar problems on several occasions, so it seems worthwhile to program a Solver-like function in VBA. We will write this in such a way that it can easily be applied in other contexts. We will assume there is a single variable for which we want to solve, solving for multiple variables being more difficult.

Letting $\tt C$ denote the market price of a European call option, the implied volatility is $\tt sigma$ satisfying

 $Black_Scholes_Call(S,K,r,sigma,q,T) - C = 0.$

The solution of this equation is called a "root" of the function

Black_Scholes_Call(S,K,r,sigma,q,T) - C,

and the problem of finding roots of functions is a standard numerical problem. Roots are found by what are essentially sophisticated trial-and-error methods. The simplest method is to start with upper and lower bounds for σ and repeatedly bisect the interval containing σ , each time finding a new upper or lower bound. The program below is a standard bisection routine.

For there to be a volatility that equates the market price to the Black-Scholes price, it is necessary for the call option price to satisfy the arbitrage bound⁵ $C + e^{-rT}K \ge e^{-qT}S$. We check this condition at the beginning of the program and supply an error message if it is violated.

We need to input all of the inputs of Black_Scholes_Call other than σ , and we need to input the call option price. The following uses an error tolerance of 10^{-6} . Therefore, the value that is returned will equal the exact implied volatility to at least five decimal places. The bisection is begun with a lower bound of $\sigma = 0$. An iterative procedure is used to find an upper bound, starting with $\sigma = 100\%$.

⁵ Note that by put-call parity—equation (3.7)—the difference between the left and right-hand sides of this inequality is the value of the put with the same strike and maturity as the call. Thus, the inequality is equivalent to the statement that the put value is nonnegative, which must be the case.

This same algorithm can be used to find a real number x such that f(x) = 0 for any (continuous) function f. The only changes necessary are in the right hand sides of the assignment statements for **flower**, **fupper**, and **fguess** and in finding lower and upper bounds (and obviously one would not check the arbitrage bound in general).⁶ We will use this algorithm on several occasions to find roots of functions.

```
Function Black_Scholes_Call_Implied_Vol(S, K, r, q, T, CallPrice)
 Inputs are S = initial stock price
             K = strike price
             r = risk-free rate
             q = dividend yield
             T = time to maturity
             CallPrice = call price
Dim tol, lower, flower, upper, fupper, guess, fguess
If CallPrice \langle Exp(-q * T) * S - Exp(-r * T) * K Then
    MsgBox ("Option price violates the arbitrage bound.")
    Exit Function
End If
tol = 10 ^ -6
lower = 0
flower = Black_Scholes_Call(S, K, r, lower, q, T) - CallPrice
upper = 1
fupper = Black_Scholes_Call(S, K, r, upper, q, T) - CallPrice
                          ' double upper until it is an upper bound
Do While fupper < 0
    upper = 2 * upper
    fupper = Black_Scholes_Call(S, K, r, upper, q, T) - CallPrice
Loop
guess = 0.5 * 1 \text{ower} + 0.5 * \text{upper}
fguess = Black_Scholes_Call(S, K, r, guess, q, T) - CallPrice
Do While upper - lower > tol ' until root is bracketed within tol
    If fupper * fguess < 0 Then ' root is between guess and upper
                                  ' make guess the new lower bound
        lower = guess
        flower = fguess
        guess = 0.5 * lower + 0.5 * upper ' new guess = bi-section
        fguess = Black_Scholes_Call(S,K,r,guess,q,T) - CallPrice
    Else
                                  ' root is between lower and guess
                                  ' make guess the new upper bound
        upper = guess
        fupper = fguess
        guess = 0.5 * lower + 0.5 * upper ' new guess = bi-section
```

 $^{^{6}}$ The key to the function is checking each time whether the root is between the guess and the upper bound or between the guess and the lower bound. If fupper \times fguess < 0, then there is a root between the guess and the upper bound. In this case, we define the new lower bound to be the old guess and define the new guess to be the midpoint of this new lower bound and the old upper bound. We do the opposite if we find the root is between the guess and the lower bound.

```
fguess = Black_Scholes_Call(S,K,r,guess,q,T) - CallPrice
End If
Loop
Black_Scholes_Call_Implied_Vol = guess
End Function
```

To compute an implied volatility from a put option price, one can first compute a corresponding call option price from put-call parity and then run the above program.

There are faster root-finding methods than bisection. These use other methods to update the guess than just halving the distance between the prior guess and the upper or lower bound. For example, one can use the vega (the derivative of the option formula with respect to σ) at the given guess for σ and replace the bisection with

```
guess = guess - call/vega.
```

This amounts to approximating the Black-Scholes formula as being linear in σ and using the root of the approximation as the updated guess. This is the essence of the *Newton-Raphson* method. A similar idea that does not require the computation of vega is to keep track of the two most recent (guess, call) pairs and to approximate vega as:

vega = (call - prior_call) / (guess - prior_guess).

This is the essence of the *secant* method.

Discretely-Rebalanced Delta Hedges

To compute the real-world distribution of gains and losses from a discretelyrebalanced delta hedge, we input the expected rate of return μ . We consider adjusting the hedge at dates $0 = t_0 < t_1 < \cdots < t_N = T$, with $t_i - t_{i-1} = \Delta t = T/N$ for each *i*. The changes in the natural logarithm of the stock price between successive dates t_{i-1} and t_i are simulated as

$$\Delta \log S = \left(\mu - q - \frac{1}{2}\sigma^2\right) \Delta t + \sigma \Delta B ,$$

where ΔB is normally distributed with mean zero and variance Δt . The random variables ΔB are simulated as standard normals multiplied by $\sqrt{\Delta t}$. We begin with the portfolio that is short a call, long δ shares of the underlying, and short $\delta S - C$ in cash. After the stock price changes, say from S to S', we compute the new delta δ' . The cash flow from adjusting the hedge is $(\delta - \delta')S'$. Accumulation (or payment) of interest on the cash position is captured by the factor $e^{r\Delta t}$. Continuous payment of dividends is modelled similarly: the dividends earned during the period Δt is taken to be $\delta S (e^{q\Delta t} - 1)$. The cash position is adjusted due to interest, dividends, and the cash flow from adjusting the hedge. At date T, the value of the portfolio is the cash position less the intrinsic value of the option.

To describe the distribution of gains and losses, we compute percentiles of the distribution. You should see that the hedge becomes more nearly perfect as the number of periods N is increased. Note that this is true regardless of the μ that is input, which reaffirms the point that option values and hedges do not depend on the expected rate of return of the underlying. The percentile is calculated with the Excel Percentile function.⁷

```
Function Simulated_Delta_Hedge_Profit(S0,K,r,sigma,q,T,mu,M,N,pct)
 Inputs are SO = initial stock price
            K = strike price
             r = risk-free rate
             sigma = volatility
             q = dividend yield
             T = time to maturity
            mu = expected rate of return
             N = number of time periods
             M = number of simulations
             pct = percentile to be returned
Dim dt, SigSqrdt, drift, LogSO, CallO, DeltaO, CashO, Comp, Div
Dim S, LogS, Cash, NewS, Delta, NewDelta, HedgeValue, i, j
Dim Profit() As Double
ReDim Profit(M)
dt = T / N
SigSqrdt = sigma * Sqr(dt)
drift = (mu - q - 0.5 * sigma * sigma) * dt
Comp = Exp(r * dt)
Div = Exp(q * dt) - 1
LogSO = Log(SO)
                              ' store log of initial stock price
Call0 = Black_Scholes_Call(S0, K, r, sigma, q, T)
Delta0 = Black_Scholes_Call_Delta(S0, K, r, sigma, q, T)
Cash0 = Call0 - Delta0 * S0 ' initial cash position
For i = 0 To M
    LogS = LogSO
                              ' initialize log of stock price
    Cash = Cash0
                              ' initialize cash position
    S = S0
                              ' initialize beginning stock price
    Delta = DeltaO
                              ' initialize beginning stock position
    For j = 1 To N - 1
        LogS = LogS + drift + SigSqrdt * RandN()
                                                    ' new log S
        NewS = Exp(LogS)
                                                    ' new S
```

⁷ If numsims = 11 and pct =0.1, the percentile function returns the second lowest element in the series. The logic is that 10% of the numbers, excluding the number returned, are below the number returned—i.e., 1 out of the other 10 are below—and 90% of the others are above. In particular, if pct = 0.5, the percentile function returns the median. When necessary, the function interpolates; for example, if numsims = 10 and pct=0.1, then the number returned is an interpolation between the lowest and second lowest numbers.

```
NewDelta = Black_Scholes_Call_Delta(NewS,K,r,sigma,q,T-j*dt)
Cash = Comp*Cash + Delta*S*Div - (NewDelta-Delta)*NewS
S = NewS ' update stock price
Delta = NewDelta ' update stock position
Next j
LogS = LogS+drift+SigSqrdt*RandN() ' final log of stock price
NewS = Exp(LogS) ' final stock price
HedgeValue = Comp*Cash + Delta*S*Div + Delta*NewS
Profit(i) = HedgeValue - Application.Max(NewS-K,0)
Next i
Simulated_Delta_Hedge_Profit = Application.Percentile(Profit, pct)
End Function
```

Problems

3.1. Create an Excel worksheet in which the user inputs K, r, σ , q and T. Compute the delta of a call option for stock prices S = .01K, .02K, ..., 1.99K, 2K (i.e., S = iK/100 for i = 1, ..., 200) and plot the delta against the stock price.

3.2. The delta of a digital option that pays \$1 when S(T) > K is

$$\frac{\mathrm{e}^{-rT}\,\mathrm{n}(d_2)}{\sigma S\sqrt{T}}$$

Repeat the previous problem for the delta of this digital. Given that in reality it is costly to trade (due to commissions, the bid-ask spread and possible adverse price impacts for large trades), do you see any problems with delta hedging a short digital near maturity if it is close to being at the money?

3.3. Repeat Prob. 3.1 for the gamma of a call option.

3.4. Use put-call parity to derive the Greeks of a put option, and write a VBA function that computes the value and Greeks.

3.5. Consider delta and gamma hedging a short call option, using the underlying and a put with the same strike and maturity as the call. Calculate the position in the underlying and the put that you should take, using the analysis in Sect. 3.6. Will you ever need to adjust this hedge? Relate your result to put-call parity.

3.6. The delta of a share digital that pays one share when S(T) > K is

$$e^{-qT} N(d_1) + \frac{e^{-qT} n(d_1)}{\sigma \sqrt{T}}$$

Repeat Prob. 3.1 for the delta of this share digital.

3.7. Create an Excel worksheet in which the user inputs K, r, q and T. Compute the value of an at-the-money call option (S = K) using the function Black_Scholes_Call for volatilities $\sigma = .01, .02, ..., 1.0$. Plot the call value against the volatility.

3.8. Repeat the previous problem for S = 1.2K (an example of an in-themoney call option).

3.9. The file CBOEQuotes.txt (available at www.kerryback.net) contains price data for call options on the S&P 500 index. The options expired in February, 2003, and the prices were obtained on January 22, 2003. The first column lists various exercise prices. The second column gives the bid price and the third column the ask price. Import this data into an Excel worksheet and compute and plot the implied volatility against the exercise price using this data. Use the ask price as the market price for the option. The options have 30 days to maturity (so T = 30/365). At the time the quotes were downloaded, the S&P 500 was at 884.25. According to the CBOE, the dividend yield on the S&P 500 was 1.76%. Use 1.25% for the risk-free interest rate.

3.10. Attempt to repeat the previous problem using the bid price as the market price of the option. If this doesn't work, what is wrong? Does this indicate there is an arbitrage opportunity?

3.11. Suppose an investor invests in a portfolio with price S and constant dividend yield q. Assume the investor is charged a constant expense ratio α (which acts as a negative dividend) and at date T receives either his portfolio value or his initial investment, whichever is higher. This is similar to a popular type of variable annuity. Letting D denote the number of dollars invested in the contract, the contract pays

$$\max\left(D, \frac{De^{(q-\alpha)T}S(T)}{S(0)}\right)$$
(3.16)

at date T. We can rearrange the expression (3.16) as

$$\max\left(D, \frac{D\mathrm{e}^{(q-\alpha)T}S(T)}{S(0)}\right) = D + \max\left(0, \frac{D\mathrm{e}^{(q-\alpha)T}S(T)}{S(0)} - D\right)$$
$$= D + \mathrm{e}^{-\alpha T}D\max\left(0, \frac{\mathrm{e}^{qT}S(T)}{S(0)} - \mathrm{e}^{\alpha T}\right) . \quad (3.17)$$

Thus, the contract payoff is equivalent to the amount invested plus a certain number of call options written on the gross holding period return $e^{qT}S(T)/S(0)$. Note that $Z(t) = e^{qt}S(t)/S(0)$ is the date-t value of the portfolio that starts with 1/S(0) units of the asset (i.e., with a \$1 investment) and reinvests dividends. Thus, the call options are call options on a non-dividend paying portfolio with the same volatility as S and initial price of \$1. This implies that the date-0 value of the contract to the investor is $e^{-rT}D$ plus Exp(-alpha*T)*D*Black_Scholes_Call(1,Exp(alpha*T),r,sigma,0,T)

(a) Create a VBA function to compute the fair expense ratio; i.e., find α such that the date-0 value of the contract is equal to D. Hint: Modify the

Black_Scholes_Call_Implied_Vol

function. You can use $\alpha = 0$ as a lower bound. Because the value of the contract is decreasing as α increases, you can find an upper bound by iterating until the value of the contract is less than D.

(b) How does the fair expense ratio vary with the maturity T? Why?

3.12. Modify the function Simulated_Delta_Hedge_Profit to compute percentiles of gains and losses for an investor who writes a call option and constructs a delta and gamma hedge using the underlying asset and another call option. Include the exercise price of the call option used to hedge as an input, and assume it has the same time to maturity as the option that is written. Hint: In each period j = 1 to N-1, the updated cash position can be calculated as

where **a** denotes the number of shares of the stock held, **b** denotes the number of units held of the option that is used for hedging, and **PriceHedge** denotes the price of the option used for hedging (computed from the Black-Scholes formula each period). This expression embodies the interest earned (paid) on the cash position, the dividends received on the shares of stock and the cash inflows (outflows) from adjusting the hedge. At the final date N, the value of the hedge is

and the value of the overall portfolio is the value of the hedge less

Application.Max(NewS-KWritten,0) ,

where KHedge denotes the strike price of the option used to hedge and KWritten denotes the strike of the option that was written.